

Supplementary File 1. Data example

SAS syntax that creates a dataset that approximates the one published in Hammond et al. [8]. Outcome is lung cancer deaths per 100,000 workers. The prevalence of smoking is assumed to be 0.28 for both the asbestos workers and for the comparison group of unexposed workers.

```
/*Data example*/
data one;
  array asbn (2) (73763 17800); /*n in published study*/
  array rate (4) (11.3 122.6 58.4 601.6);
                                /*lung ca deaths per 100k in published study*/
  smokeprev=0.28; /*assumed prevalence of smoking*/
  do asbestos=0 to 1;
    do smk=0 to 1;
      do lungcadeath=0 to 1;
        mult=rate [2*asbestos + smk +1] / 100000;
        count1=asbn[asbestos+1] *
                (abs((1-smk)-smokeprev)) *
                (abs((1-lungcadeath)-mult));
        count=round(count1,1);
        output;
      end;
    end;
  end;
  keep asbestos smk lungcadeath count;
run;
proc sort data=one (keep=asbestos smk lungcadeath count) out=two;
  by descending asbestos descending smk descending lungcadeath;
run;

/*version of dataset with individual observations*/
data long;
  set two;
  do i=1 to count;
    id+1;
    output;
  end;
run;

proc format;
  value smkf 1="Smokers" 0="Non-smokers";
  value gpf 1="Asbestos Workers (n= 17800)"
            0="Comparison Group (n=73763)";
  value death 1="Deaths due to lung CA"
              0="Alive or dead due to other causes";
run;

/* Table 2. Lung cancer deaths (per 100,000 workers) among those with exposure
to asbestos and/or cigarette smoking*/
proc freq data=two order=data;
  weight count;
  tables asbestos*smk*lungcadeath / nocol nopct outpct out=three;
  format smk smkf. asbestos gpf. lungcadeath death.;
run;
```

```
data four;
  set three;
  perhunthou=pct_row*1000;
run;

proc report nowd data=four;
  where lungcadeath=1;
  columns smk asbestos, perhunthou;
  define smk / group "Cigarette smoking" format=smkf. order=data;
  define asbestos / across "Asbestos Exposure" format=gpf. order=data;
  define perhunthou / analysis '' format=6.2;
run;
```

Supplementary File 2. SAS syntax for linear binomial model

The syntax below illustrates the construction of the linear binomial model [4,7] using SAS PROC GENMOD. A MODEL statement identifies the independent variables included in comprise the linear predictor: smk (smoking status); asbestos (asbestos exposure status); and smk*asbestos, the interaction between smoking status and asbestos exposure. Options in the MODEL statement specify that the outcome (lung cancer) follows a binomial distribution and link it directly (through an identity link) to the linear predictor. The LSMEANS statement estimates the number of deaths per 100,000 workers for each combination of exposures. The ESTIMATE statement lists the four coefficients (1 -1 -1 1) that define the interaction contrast (IC):

$$(1) p_{11} + (-1)p_{10} + (-1)p_{01} + (1)p_{00} = 0$$

```
/*linear binomial model*/
proc genmod data=long descending;
  class smk (ref=first) asbestos (ref=first);
  model lungcadeath = smk asbestos smk*asbestos
    / link=identity dist=bin type3 wald;
  lsmeans smk*asbestos / cl;
  ods output lsmeans=lsmeans estimates=estimates parameterestimates=betas
modelanova=type3;
  estimate "IC" smk*asbestos 1 -1 -1 1;
run;

/*Syntax that includes a REPEATED statement, which initiates GEE estimation
of robust standard errors, advocated by Richardson et al.[7].*/
proc genmod data=long descending;
  class smk (ref=first) asbestos (ref=first) id;
  model lungcadeath = smk asbestos smk*asbestos
    / link=identity dist=bin type3 wald;
  repeated subject=id / type=ind;
  lsmeans smk*asbestos / cl;
  estimate "IC" smk*asbestos 1 -1 -1 1;
run;

/*modification of linear binomial model advocated by Spiegelman and Herzmark
[4] for instances when convergence fails*/
proc genmod data=long descending;
  class smk (ref=first) asbestos (ref=first) id;
  model lungcadeath = smk asbestos smk*asbestos
    / link=identity dist=poisson type3 wald ;
  repeated subject=id / type=ind;
  lsmeans smk*asbestos / cl;
  estimate "IC" smk*asbestos 1 -1 -1 1;
run;
```

Supplementary File 3. SAS syntax for Table 3 and Figure 1

The syntax below uses data sets output from the linear binomial model (Supplementary Box S2) to create Table 3 and Figure 1.

```
/* Table 3. Absolute risks (and risk differences) for death from lung cancer
(per 100,000 workers) for those with exposure to asbestos and/or cigarette
smoking, estimated by linear binomial model*/
data mortality;
  set lsmeans;
  mortality=estimate*100000;
  ucl=upper*100000;
  lcl=lower*100000;
run;
proc print noobs data=mortality;
  var smk asbestos estimate mortality lcl ucl;
run;

/*estimate for interaction contrast (IC)*/
data ic;
  set estimates;
  ic=meanestimate*100000;
  ic_lcl=meanlowercl*100000;
  ic_ucl=meanuppercl*100000;
run;
proc print noobs data=ic;
  var label meanestimate ic ic_lcl ic_ucl probchisq;
  format meanestimate 9.6 probchisq 12.8 ;
run;

/*Estimates of regression coefficients, which are interpretable as excess
deaths*/
data beta2;
  set betas (where=(df=1));
  excessdeaths=estimate*100000;
  ucl=upperwaldcl*100000;
  lcl=lowerwaldcl*100000;
run;
proc print noobs data=beta2;
  var parameter estimate excessdeaths lcl ucl probchisq;
  format estimate 9.6 probchisq 12.8;
run;

/* Figure 1. Biological interaction, between asbestos exposure and smoking,
illustrated as a non-additivity or heterogeneity of effects*/
proc template;
  define style mystyle;
    parent=styles.default;
    class graphbackground / color=white;
    style GraphData1 from GraphData1 /
      contrastcolor=black linestyle=1;
    style GraphData2 from GraphData2 /
      contrastcolor=black linestyle=2;
  end;
run;
```

```

ods html style=styles.mystyle;
proc sgplot data=mortality;
  series y=mortality x=smk / group=asbestos name="one"
        groupdisplay=cluster clusterwidth=0.05
        markers markerattrs=(symbol=squarefilled size=10);
  highlow x=smk high=ucl low=lcl / group=asbestos
        groupdisplay=cluster clusterwidth=0.05
        type=line lineattrs=(pattern=1) lowcap=serif highcap=serif;
  xaxis values=(0 1) label=" " valueattrs=(size=14 weight=bold);
  yaxis label="Lung cancer deaths per 100,000"
        labelattrs=(size=14 weight=bold)
        valueattrs=(size=14 weight=bold);
  format smk smkf. asbestos gpf.;
  keylegend "one" / title="" location=inside down=2 position=topleft
        valueattrs=(size=12 weight=bold) ;
run;
ods html close;

```

Supplementary File 4. Log binomial and logistic regression models

Log binomial model and depiction of its estimates in Figure 2.

```
proc genmod data=long descending;
  class smk (ref=first) asbestos (ref=first) ;
  model lungcadeath = smk asbestos smk*asbestos
    / link=log dist=bin type3 wald lrci;
  lsmeans smk*asbestos / cl;
  ods output lsmeans=lsmeans ;
run;

ods html style=styles.mystyle;
proc sgplot data=lsmeans;
  series y=estimate x=smk / group=asbestos name="one"
    groupdisplay=cluster clusterwidth=0.05
    markers markerattrs=(symbol=squarefilled size=10);
  highlow x=smk high=upper low=lower / group=asbestos
    groupdisplay=cluster clusterwidth=0.05
    type=line lineattrs=(pattern=1) lowcap=serif highcap=serif;
  xaxis values=(0 1) label=" " valueattrs=(size=14 weight=bold);
  yaxis label="ln[p(Death from lung cancer)]"
    labelattrs=(size=14 weight=bold)
    valueattrs=(size=14 weight=bold);
  format smk smkf. asbestos gpf.;
  keylegend "one" / title="" location=inside down=2 position=topleft
    valueattrs=(size=12 weight=bold) ;
run;
ods html close;
```

Logistic regression model and depiction of its estimates in Figure 3.

```
proc genmod data=long descending;
  class smk (ref=first) asbestos (ref=first) ;
  model lungcadeath = smk asbestos smk*asbestos
    / link=logit dist=bin type3 wald lrci;
  lsmeans smk*asbestos / cl;
  ods output lsmeans=lsmeans ;
run;

ods html style=styles.mystyle;
proc sgplot data=lsmeans;
  series y=estimate x=smk / group=asbestos name="one"
    groupdisplay=cluster clusterwidth=0.05
    markers markerattrs=(symbol=squarefilled size=10);
  highlow x=smk high=upper low=lower / group=asbestos
    groupdisplay=cluster clusterwidth=0.05
    type=line lineattrs=(pattern=1) lowcap=serif highcap=serif;
  xaxis values=(0 1) label=" " valueattrs=(size=14 weight=bold);
  yaxis label="Log odds of death from lung cancer"
    labelattrs=(size=14 weight=bold)
    valueattrs=(size=14 weight=bold);
  format smk smkf. asbestos gpf.;
  keylegend "one" / title="" location=inside down=2 position=topleft
```

```
valueattrs=(size=12 weight=bold) ;  
run;  
ods html close;
```